digital

DISPLAY



DEC-12-FLSB-D
FEBRUARY, 1971

The material in this handbook, including but not limited to instruction times and operating speeds, is for information purposes and is subject to change without notice.

The following are trademarks of Digital Equipment Corporation, Maynard, Massachusetts:

DEC	PDP
FLIP CHIP	FOCAL
DIGITAL	COMPUTER LAB

The equipment described herein is covered by patents and patents pending.

For additional copies order DEC-12-FLSB-D from Program Library, Digital Equipment Corporation, 146 Main Street, Maynard, Massachusetts 01754 Price $5.00

CONTENTS

DISPLAY

1. INTRODUCTION

DISPLAY enables a data display facility for those routines which do not require complex display processing or cannot sacrifice the core for such a display. The routine displays any contiguous section of core via a moving window, with a cursor and octal read out of cursor positions to facilitate operator interaction.

2. ENVIRONMENT

2.1 Hardware

A PDP-12A computer.

2.2 Core Allocation

Any two PDP-8 pages in field Ø except page Ø that are segment contained, beta registers Ø, 1, and 2 of the segment, and any six page Ø locations are the core requirements. The actual binary output resides in one tape block.

3. CALLING PROCEDURE

There are two calls to DISPLAY: an initial call and a refresh call, both of which must be in the same PDP-8 field as DISPLAY (field Ø). The function of the initial call is to set up arguments for the refresh subroutines:

```
JMS I KIDORA

FIELD
CORE LOCATION
FIELD
CORE LOCATION
Y OFFSET
Y SCALE FACTOR
    .
    .
    .
    .
KIDORA, IDORA
```

The first four words are two 15-bit addresses that specify the beginning and end of the data buffer to be displayed, respectively. Any contiguous section of core is legal input to IDORA as long as the buffer is at least $1ØØØ_8$ points long. The first address must be less than the second. DISPLAY treats the data buffer as a core ring so that the initial and terminal points are contiguously displayed.

The purpose of words 5 and 6 is to allow the calling programs flexibility in displaying data.[1] The fifth word is a Y offset and specifies a 12 bit number to be added (2's compliment) to the data before it is displayed; the sixth word is a SCR K instruction and indicates the amount of scaling to be done prior to display. IDORA exits to the SCR K location in PDP-8 mode with the data field set to Ø and the AC = Ø.

The refresh call is simply

```
JMS I KRDORA
  .
  .
  .
KRDORA, RDORA
```

The refresh call displays $1ØØØ_8$ points, sets arguments for the next refresh and returns to the location following the call, in PDP-8 mode with the accumulator cleared and the data field DF and IF unchanged. Note that the initial call to DISPLAY must be to IDORA; RDORA always refreshes the buffer specified by the last call to IDORA

[1] It should be noted that DISPLAY assumes the data to be in single word format.

4. FACTOR INTERACTION

The operator controls the position of the window with knob Ø: clockwise motion moves the window to the "right" or towards the end of the buffer; counter-clockwise, to the left. The midpoint reading on knob Ø causes the motion to stop.

Knobs 1 and 5 and Sense Switch (SSW) 5 control the cursor (an intensified dot). Depending on the setting of SSW 5, the cursor may either move along the curve or be displayed independently. When SSW 5 = Ø, the cursor moves along the curve and its position is controlled with Knob 1:

When Knob 1 is turned to its furthermost clockwise position, the cursor sits upon the rightmost scope point; when Knob 1 is positioned to its furthermost counter-clockwise position, the cursor sits on the leftmost scope point; intervening knob positions yield intervening cursor positions.

When SSW5 = 1, the cursor is displayed independently of the curve. For this case, Knob 1 controls the horizontal coordinate and Knob 5 the vertical coordinate. Horizontal displacement of the cursor via Knob 1 is identical to that described above. When Knob 5 is at its furthermost clockwise position, the cursor is displayed at the top of the scope. When Knob 5 is at its furthermost counter-clockwise position, the cursor is displayed at the bottom of the scope.

Associated with the cursor are four octal words displayed in the top left corner of the scope, one beneath the other[1]. The first two words are the absolute 15-bit core address of the cursor point. The third word is the contents of the displayed core address, i.e., the actual 12-bit value in the data buffer of the data word that corresponds to the cursor point. The fourth word is the scope Y coordinate of the cursor point. The fourth word is a relative value and depends upon the Y scale factor and Y offset. If the data had been scaled to nine bits prior to display, the fourth word or Y coordinate would range from ØØØ1 to $1ØØØ_8$, where ØØØ1 corresponds to the bottom of the scope and 1ØØØ corresponds to the top.

To facilitate interaction with the calling program the four displayed words described above are maintained in page Ø and may be accessed after the refresh return.

TAG	CONTENTS
XCURHI	Fifteen bit address of the point
XCURLO	in memory reference by the cursor
CORVAL	Contents of memory
YCUR	Relative Y display coordinate

[1]The character size of the display depends upon the setting of the special functions register at the time the display is refreshed.

Standard Parameters

The six locations on page Ø can be anywhere and they do not have to be contiguous. The standard version of DISPLAY assumes locations 2Ø - 25; the locations are defined via direct assignment statements at the end of the DISPLAY source.

```
YSHFT = 2Ø                  CONTAINS Y OFFSET
XCURHI = YSHFT + 1
XCURLO = XCURHI + 1
CORVAL = XCURLO + 1         SEE ABOVE
YCUR = CORVAL + 1
COUNT = YCUR + 1
```

The following symbols (also defined at the end of the DISPLAY source) may be redefined if the user wishes to use Knobs other than Ø, 1, and 5:

```
WINSAM = SAM Ø
CURSAM = SAM 1
FRESAM = SAM 5
```

As coded, DISPLAY occupies locations 54ØØ through 5777, with IDORA = 54ØØ and RDORA = 5544. The origin statement at line 1Ø and/or the PAGE pseudo-op at line 223 may be altered to relocate part or all of DISPLAY. (N.B. if using the page of the segment containing the beta registers, place the origin at the fourth location of the page since DISPLAY uses beta registers Ø,1, and 2; i.e. * 2ØØ3; or * 4ØØ3; or * 6ØØ3).

Example

If Segments 4 and 5 contain 12 bit, unsigned data, the following would display 512 points (as controlled by knob Ø) until a keyboard input is detected:

```
        JMS I KIDORA        /N.B.  PDP-8 MODE
          1                 /LOCATION 1Ø,ØØØ.
          Ø                 /START OF BUFFER
          1                 /LOCATION 13,777 IS
       3777                 /END OF BUFFER
       4ØØØ                 /VERTICAL COORDINATOR
                            /377      1ØØØ
       SCALE 3              /  Ø  TO   4ØØ
       JMS I KRDORA         /4ØØ        1
       KSF                  /FROM 12 BITS TO 9
       JMP LOOP             /REFRESH SCOPE
          .                 /KEY STRUCK ?
          .                 /NO, CONTINUE DISPLAY
          .
          .                 The program might now do some-
                            thing with the datum indicated
                            by the cursor   exit, or .....

    KIDORA,IDORA            /INITIALISE POINTER
    KRDORA,RDORA            /REFRESH POINTER
```

5. INTERNAL DESCRIPTION

5.1 IDORA

The function of IDORA is to retrieve the arguments from the initial call and set them up for RDORA. The lower bound is stored at MINFLD. MINADR, the upper bound + 1, is stored at MAXFLD, MAXADR. The Y offset is stored at YSHFT (page Ø) and the scale factor is stored at YSCAL.

5.2 RDORA

The function of RDORA is to display $1ØØØ_8$ points relative to the window knob and a scope starting address and return. RDORA first reads the horizontal cursor knob and sets it to a value between -1 and $-1ØØØ_8$; this is the cursor count. RDORA then reads the window knob and updates the initial scope address (BUFHI, BUFLO) maintaining that address modulo the upper and lower bounds. RDORA then displays the buffer one point at a time. After each point is displayed, RDORA checks (ISZ) the cursor count; when it goes to Ø, the cursor is displayed. RDORA also checks for the end of the display buffer and the end of a field, resetting to the start of the display buffer or to the next field respectively, as the case may be. After all points have been displayed, the routine DSCLOC is called to display the four words in the upper left corner of the scope. RDORA then restores the user's data field and returns.

5.3 FLOW DIAGRAM






RDORA
SAVE USER DATA FIELD
READ HORIZ CURSOR KNOB WINDOW KNOB
ADD WINDOW KNOB TO SCOPE ADDR
LOWER BOUND EXCEEDED ?
Y
N
UPPER BOUND EXCEEDED ?
Y
N
RESET SCOPE ADDR.
SETUP DISPLAY ARGS
NXTPNT


NXTPNT
DISP 1 PNT
DISP CURSOR ?
Y
CALL CURDIS
N
Y
AT UPPER BOUND ?
Y
RESET TO LOWER BOUND
N
FIELD BOUNDARY ?
Y
SET NXT DF
N
DISP 512 PNTS ?
Y
N
DSCLOC
NXTPNT


CURDIS
SAVE X,Y COORDS
FREE STANDING MODE ?
Y
SET Y COORD TO VERT. KNOB
N
DISP CURSOR 1 TIME
DONE ?
N
Y
RTN

5.4 LISTING

```
0000                    *20
0001                    /DISPLAY (MINI MAGSPY )
0002                    /COPYRIGHT 1970
0003                    /DIGITAL EQUIPMENT CORPORATION,MAYNARD,MASS.
0004                    /MOVING WINDOW DISPLAY SUBROUTINE
0005                    /CURSOR READ OUT
0006                    /CORE OR TAPE FILE
0007                            PMODE
0010                            *5400
0011
0012
0013                    /INITIAL CALL GET PARAM LIST
0014                    /SET ARGS FOR RDORA
0015
0016
0017    5400    0000    IDORA,  0               /GET BOUNDS
0020    5401    7300            CLA CLL
0021    5402    6201    ACDF0,  CDF 0
0022    5403    1600            TAD I IDORA     /DATA BUFFER
0023    5404    3635            DCA I KMNFLD    /15 BIT
0024    5405    2200            ISZ IDORA       /LOWER BOUND
0025    5406    1600            TAD I IDORA     /AT P+1, P+2
0026    5407    3636            DCA I KMNADR    /MINFLD,MINADR
0027    5410    2200            ISZ IDORA
0030    5411    1600            TAD I IDORA     /UPPER BOUND
0031    5412    3637            DCA I KMXFLD    /AT P+3, P+4
0032    5413    2200            ISZ IDORA
0033    5414    7001            IAC             /RDORA USES
0034    5415    1600            TAD I IDORA     /MAX+1
0035    5416    3640            DCA I KMXADR
0036    5417    7004            RAL
0037    5420    1637            TAD I KMXFLD
0040    5421    3637            DCA I KMXFLD
0041    5422    2200            ISZ IDORA
0042    5423    1600            TAD I IDORA     /Y SHIFT
0043    5424    3020            DCA YSHFT
0044    5425    2200            ISZ IDORA
0045    5426    1600            TAD I IDORA     /Y SCALE
0046    5427    3643            DCA I KYSCAL
0047    5430    1635            TAD I KMNFLD    /INITIALIZE
0050    5431    3641            DCA I KBUFHI    /WINDOW
0051    5432    1636            TAD I KMNADR    /STARTING ADDR
0052    5433    3642            DCA I KBUFLO
0053    5434    5600            JMP I IDORA     /RTN TO SCR N
0054    5435    5615    KMNFLD, MINFLD
0055    5436    5616    KMNADR, MINADR
0056    5437    5674    KMXFLD, MAXFLD
0057    5440    5675    KMXADR, MAXADR
0060    5441    5774    KBUFHI, BUFHI
0061    5442    5775    KBUFLO, BUFLO
0062    5443    5644    KYSCAL, YSCAL
0063    5444    0401    P401,   401
```

```
0064    5445  1244  DSCLOC, TAD P401        /DSC X,Y COORD
0065    5446  3275          DCA VCOORD
0066    5447  1021          TAD XCURHI       /FIELD
0067    5450  4262          JMS DSCWD
0070    5451  1022          TAD XCURLO       /ADDRESS
0071    5452  4262          JMS DSCWD
0072    5453  1023          TAD CORVAL       /CONTENTS OF
0073    5454  4262          JMS DSCWD        /CURSR CORE LOC
0074    5455  1024          TAD YCUR         /Y COORD OF
0075    5456  1244          TAD P401
0076    5457  4262          JMS DSCWD        /CURSOR POINT
0077    5460  0000  RTNCDF, 0                /RESTORE USER
0100                                         /DATA FLD
0101    5461  5744          JMP I RDORA      /RTN
0102    5462  0000  DSCWD,  0                /DSC C(AC)
0103    5463  6141          LINC
0104                        LMODE
0105    1464  5477          STC TEMP         /SAVE VALUE
0106    1465  4001          STC XCORD        /CHAN 1
0107    1466  0024          SPA              /VC FOR FULL
0110    1467  0265          ROL I 5          /SIZE IS -40
0111    1470  1020          LDA I            /-20 FOR HALF
0112    1471  7757          -20
0113    1472  0452          LZE              /FULL CHARS ?
0114    1473  0241          ROL 1            /NO VC-40
0115    1474  1160          ADM I            /UPDATE VC
0116    1475  0000  VCOORD, 0
0117    1476  1020  DSCLOP, LDA I
0120    1477  0000  TEMP,   0
0121    1500  0243          ROL 3            /1 DIGIT
0122    1501  1040          STA              /AT A TIME
0123    1502  1477          TEMP             /UPDATE
0124    1503  1560          BCL I            /LOW 3 BITS
0125    1504  7770          7770             /ONLY
0126    1505  0241          ROL 1            /*2 AND REL
0127    1506  1120          ADA I            /TO GRID TAB
0130    1507  1524          TAB&1777
0131    1510  4002          STC 2
0132    1511  3475          ADD VCOORD
0133    1512  1742          DSC 2
0134    1513  1762          DSC I 2
0135    1514  0221          XSK I 1          /MAKE GAP
0136    1515  0221          XSK I 1          /BETWEEN CHARS
0137    1516  1520          SRO I            /DSC 4 CHARS ?
0140    1517  3567          3567
0141    1520  7476          JMP DSCLOP       /NO CONT
0142    1521  0002          PDP
0143                        PMODE
0144    5522  7300          CLA CLL
0145    5523  5662          JMP I DSCWD      /RTN
0146    5524  4536  TAB,    4536     /60,0
0147    5525  3651          3651
0150    5526  2101          2101     /61,1
*
```

```
0151    5527  0177          0177
0152    5530  4523          4523     /62,2
0153    5531  2151          2151
0154    5532  4122          4122     /63,3
0155    5533  2651          2651
0156    5534  2414          2414     /64,4
0157    5535  0477          0477
0160    5536  5172          5172     /65,5
0161    5537  0651          0651
0162    5540  1506          1506     /66,6
0163    5541  4225          4225
0164    5542  4443          4443     /67,7
0165    5543  6050          6050
0166    5544  0000  RDORA,  0
0167    5545  7300          CLA CLL          /SAVE USER DF
0170    5546  6214          RDF
0171    5547  1202          TAD ACDF0
0172    5550  3260          DCA RTNCDF
0173    5551  6141          LINC
0174                        LMODE
0175    1552  0101  CSAM,   CURSAM           /CURSOR
0176    1553  0341          SCR 1            /9 BITS COVERS
0177    1554  0002          PDP              /SCOPE
0200                        PMODE            /MAKE RANGE
0201    5555  1244          TAD P401         /-1 TO -1000
0202    5556  7141          CIA CLL
0203    5557  6141          LINC
0204                        LMODE
0205    1560  5773          STC CURCNT&1777
0206    1561  0100  WSAM,   WINSAM           /WINDOW
0207    1562  0344          SCR 4            /75 CENTS WORTH
0210    1563  0061          SET I XCORD
0211    1564  6777          -1000            /512 PNTS
0212    1565  7600          JMP CONT&1777
0213    1566  0105  FREE,   FRESAM
0214    1567  0341          SCR 1
0215    1570  0002          PDP
0216                        PMODE
0217    5571  3024          DCA YCUR
0220    5572  1024          TAD YCUR
0221    5573  6141          LINC
0222    5574  6000          6000             /JMP 0
0223                        PAGE
0224    5600  0002  CONT,   2
0225    5601  6201  CDF0,   CDF 0
0226    5602  3347          DCA DBLLO        /PUT KNOB VAL
0227    5603  1347          TAD DBLLO        /IN DAC
0230    5604  7710          SPA CLA          /PROPAGATE SIGN
0231    5605  7040          CMA              /BIT HI ORD
0232    5606  3341          DCA DBLHI
0233    5607  4304          JMS DADD
0234    5610  1347          TAD DBLLO        /UPDATE WIN ADDR
0235    5611  3375          DCA BUFLO
```

```
0236    5612  1341          TAD DBLHI
0237    5613  3374          DCA BUFHI
0240                                      /MUST CHK
0241                                      /WINDOW SA
0242                                      /WITH BOUNDS
0243                                      /TO MAINTAIN
0244                                      /BUFFER RING
0245
0246    5614  4316          JMS BOUND     /LOWER BOUND
0247    5615  0001  MINFLD, 1
0250    5616  0000  MINADR, 0
0251    5617  7700          SMA CLA       /LOW END WRAP?
0252    5620  5273          JMP CHKHI     /NO
0253    5621  1274          TAD MAXFLD    /RESET TO
0254    5622  3374          DCA BUFHI     /UPPER BOUND
0255    5623  1275          TAD MAXADR
0256    5624  3375  WRAP,   DCA BUFLO
0257    5625  4304          JMS DADD      /CORRECT WRAP
0260    5626  1347          TAD DBLLO     /CORRECTED
0261    5627  3375          DCA BUFLO     /WINDOW SA
0262    5630  1341          TAD DBLHI
0263    5631  3374          DCA BUFHI
0264    5632  1375  SETFLD, TAD BUFLO     /SET DISPLAY
0265    5633  3304          DCA BUFPTR    /ARGS
0266    5634  1327          TAD M1000
0267    5635  3025          DCA COUNT
0270    5636  1374          TAD BUFHI
0271    5637  3316          DCA BOUND
0272    5640  4341          JMS SETDF
0273    5641  1704  NXTPNT, TAD I BUFPTR
0274    5642  1020          TAD YSHFT     /OFF SET
0275    5643  6141          LINC
0276                        LMODE
0277    1644  0341  YSCAL,  SCR 1         /SCALE FACTOR
0300    1645  0161          DIS I XCORD
0301    1646  0002          PDP
0302                        PMODE
0303    5647  2373          ISZ CURCNT    /READY TO DIS
0304                                      /CURSOR ?
0305    5650  7610  CURRTN, SKP CLA       /NO
0306    5651  5351          JMP CURDIS
0307    5652  2376          ISZ ENDLO     /CHK FOR HI
0310    5653  5263          JMP OKEND     /END WRAP
0311    5654  2377          ISZ ENDHI
0312    5655  5263          JMP OKEND
0313    5656  1216          TAD MINADR    /RESET TO
0314    5657  3304          DCA BUFPTR    /LOWER BOUND
0315    5660  1215          TAD MINFLD
0316    5661  3316          DCA BOUND
0317    5662  5256          JMP NXTDF
0320    5663  2304  OKEND,  ISZ BUFPTR    /CHK FOR FIELD
0321                                      /BOUNDARY
0322    5664  5267          JMP OKFLD     /ITS OK
```

```
0323    5665  2316          ISZ BOUND       /SET NXT FLD
0324    5666  4341  NXTDF,  JMS SETDF
0325    5667  2025  OKFLD,  ISZ COUNT       /512 PNTS ?
0326    5670  5241          JMP NXTPNT      /NO
0327    5671  5672          JMP I .+1       /DSC READ OUT
0330    5672  5445          DSCLOC
0331    5673  4316  CHKHI,  JMS BOUND       /CHK UPR BOUND
0332    5674  0002  MAXFLD, 2
0333    5675  0000  MAXADR, 0
0334    5676  7710  M70,    SPA CLA         /HI WRAP ?
0335    5677  5232          JMP SETFLD
0336    5700  1215          TAD MINFLD      /YES
0337    5701  3374          DCA BUFHI       /RESET TO
0340    5702  1216          TAD MINADR      /LOWER BOUND
0341    5703  5224          JMP WRAP
0342                /DOUBLE PRECISION ADD
0343                /(DBLHI,DBLLO)+(BUFHI,BUFLO)
0344                /RESULT IN (DBLHI,DBLLO)
0345                /(BUFHI,BUFLO)=INITIAL SCOPE ADDRESS
0346
0347    5704  0000  DADD,   0
0350    5705  7300          CLA CLL
0351    5706  1347          TAD DBLLO
0352    5707  1375          TAD BUFLO
0353    5710  3347          DCA DBLLO
0354    5711  7004          RAL
0355    5712  1341          TAD DBLHI
0356    5713  1374          TAD BUFHI
0357    5714  3341          DCA DBLHI
0360    5715  5704          JMP I DADD
0361
0362                /ADD -UPPER OR -LOWER BOUND
0363                /TO (BUFHI,BUFLO)
0364                /BOUND IS AT P+1,P+2 OF CALL
0365
0366    5716  0000  BOUND,  0
0367    5717  1716          TAD I BOUND     /2S COM OF ARG
0370    5720  7140          CMA CLL         /TO DAC
0371    5721  3341          DCA DBLHI
0372    5722  2316          ISZ BOUND
0373    5723  1716          TAD I BOUND
0374    5724  7041          CIA
0375    5725  7430          SZL
0376    5726  2341          ISZ DBLHI
0377    5727  7000  M1000,  NOP
0400    5730  3347          DCA DBLLO
0401    5731  4304          JMS DADD
0402    5732  1341          TAD DBLHI
0403    5733  3377          DCA ENDHI       /DAC HOLDS -NUM
0404    5734  1347          TAD DBLLO       /TO END OF BUF
0405    5735  3376          DCA ENDLO       /NO MATTER FOR
0406                                        /LOW END WRAP
0407    5736  1341          TAD DBLHI       /TO CHK FOR
```

```
0410    5737  2316          ISZ BOUND         /UPON RTN
0411    5740  5716          JMP I BOUND
0412    5741  0000  SETDF,  0                 /SET B FIELD
0413    5742  1316          TAD BOUND         /REL TO BOUND
0414    5743  7106          CLL RTL
0415    5744  7004          RAL
0416    5745  1201          TAD CDF0
0417    5746  3347          DCA .+1
0420    5747  0000  DBLLO,  0
0421    5750  5741          JMP I SETDF
0422    5751  3024  CURDIS, DCA YCUR          /DISP CURSOR
0423    5752  1316          TAD BOUND         /SAVE X,Y
0424    5753  3021          DCA XCURHI        /COORDINATES
0425    5754  1304          TAD BUFPTR
0426    5755  3022          DCA XCURLO
0427    5756  1704          TAD I BUFPTR
0430    5757  3023          DCA CORVAL
0431    5760  1276          TAD M70
0432    5761  3347          DCA DBLLO
0433    5762  1024          TAD YCUR
0434    5763  6141  CURLOP, LINC
0435                        LMODE
0436    1764  0465          SNS I 5
0437    1765  7566          JMP FREE          /FREE CURSOR
0440    1766  0141          DIS XCORD
0441    1767  0002          PDP
0442                        PMODE
0443    5770  2347          ISZ DBLLO
0444    5771  5363          JMP CURLOP
0445    5772  5250          JMP CURRTN
0446    5773  0000  CURCNT, 0
0447                /THESE 5 GUYS MAY BE PAGE 0
0450    5774  0001  BUFHI,  1
0451    5775  0000  BUFLO,  0
0452    5776  0000  ENDLO,  0
0453    5777  0000  ENDHI,  0
0454                        DBLHI=SETDF
0455                        BUFPTR=DADD
0456                        XCORD=1
0457                        LMODE
0460                        CURSAM=SAM 1      /CURSOR KNOB
0461                        WINSAM=SAM 0      /WINDOW KNOB
0462                        FRESAM=SAM 5      /FREE CURSOR
0463                        SCALE=SCR
0464                        SC12BU=SCR 3      /SCALE FACTOR
0465                                          /12 BIT UNSIGNED
0466                        OF12BU=4000       /Y OFFSET FOR
0467                                          /12 BIT UNSIGNED
0470                /THESE 6 GUYS MUST BE PAGE 0
0471                /THEY ARE ALL CONTIGUOUS AND DEFINED
0472                /RELATIVE TO YSHFT BUT THIS IS NOT
0473                /A REQUIREMENT
0474                        YSHFT=20
=
```

```
0475                    /THE 4 SCOPE READ OUT VALUES
0476                            XCURHI=YSHFT+1
0477                            XCURLO=XCURHI+1
0500                            CORVAL=XCURLO+1
0501                            YCUR=CORVAL+1
0502                            COUNT=YCUR+1
0503                    /AN EXAMPLE TO DISPLAY ALL OF FIELD 1
0504                    /12 BIT UNSIGNED DATA
0505                    /PMODE
0506                    /*SOMEWHERE
0507                    /JMS I KIDORA           /INITIAL CALL
0510                    /1                      /FLD
0511                    /0000                   /ADDR
0512                    /1                      /FLD
0513                    /7777                   /ADDR
0514                    /OF12BU                 /Y OFF SET
0515                    /SC12BU                 /Y SCALE FAC
0516                    /JMS I KRDORA           /REFRESH CALL
0517                    /JMP .-1
0520                    /KIDORA,IDORA
0521                    /KRDORA,RDORA
0522
0523                    /DISPLAY 3 AUG 70
```

NO ERRORS

```
ACDF0  5402
BOUND  5716
BUFHI  5774
BUFLO  5775
BUFPTR 5704
CDF0   5601
CHKHI  5673
CONT   5600
CORVAL 0023
COUNT  0025
CSAM   5552
CURCNT 5773
CURDIS 5751
CURLOP 5763
CURRTN 5650
CURSAM 0101
DADD   5704
DBLHI  5741
DBLLO  5747
DSCLOC 5445
DSCLOP 5476
DSCWD  5462
ENDHI  5777
ENDLO  5776
FREE   5566
FRESAM 0105
IDCRA  5400
KBUFHI 5441
KBUFLO 5442
KMNADR 5436
KMNFLD 5435
KMXADR 5440
KMXFLD 5437
KYSCAL 5443
-
```

```
ACDF0  5402
BOUND  5716
BUFHI  5774
BUFLO  5775
BUFPTR 5704
CDF0   5601
CHKHI  5673
CONT   5600
CORVAL 0023
COUNT  0025
CSAM   5552
CURCNT 5773
CURDIS 5751
CURLOP 5763
CURRTN 5650
CURSAM 0101
DADD   5704
DBLHI  5741
DBLLO  5747
DSCLOC 5445
DSCLOP 5476
DSCWD  5462
ENDHI  5777
ENDLO  5776
FREE   5566
FRESAM 0105
IDORA  5400
KBUFHI 5441
KBUFLO 5442
KMNADR 5436
KMNFLD 5435
KMXADR 5440
KMXFLD 5437
KYSCAL 5443
```